# delimit ;
clear ;
set more off ;
est drop _all ;
cd "R:/personlig/fenellac/menarche_replication/analysis/" ;

global plusdir "R:/personlig/fenellac/stata_ado/plus/" ;
sysdir set PLUS $plusdir ;
adopath ++ $plusdir ;

* **************************************************************************** ;
* This code replicates Appendix F, Table 1.
*
* Paper: "Age of Marriage and Women's Political Engagement: Evidence from India"
* Authors: Fenella Carpena, Francesca Jensenius
* 
* Code by Fenella Carpena
* Last update: March 3, 2020
* **************************************************************************** ;

****************************************************************************** ;
* select the sample
****************************************************************************** ;

use "./input/eligible-women-ihds02-lhs.dta", clear ;
merge 1:1 IDHH PERSONID using "./input/eligible-women-ihds02-rhs.dta" ;
tab _merge ;
assert _merge == 3 ;
drop _merge ;

* keep only rural women ;
keep if rural == 1 ;

* keep only women who are married only once (i.e., have not ever remarried) ;
drop if ever_remarried == 1 ;
  
* keep only women age of menarche between 11-18 (1st-99th pctile) ; 
summ menarche_age ;
keep if menarche_age >= 11 & menarche_age <= 18 ;

* generating district identifier ;
egen dt_id = group(STATEID DISTID) ;

****************************************************************************** ;
* label variables for the output table 
****************************************************************************** ;

local political "attended_panchayat discuss_politics member_pol_org" ;

label var attended "\shortstack[l]{Attended village \\ council meeting \\ last year}" ;
label var discuss_politics "\shortstack[l]{Discusses politics \\ and community \\ with husband}" ;
label var member_pol_org "\shortstack[l]{Participates in \\ a political \\ organization}" ;

label var marriage_age "Marriage Age" ;

****************************************************************************** ;
* TABLE: OLS regression comparing the political behavior of married vs. no longer married women ;
****************************************************************************** ;
est drop _all ;

foreach var of varlist `political' { ;
 	areg `var' menarche_age height_measure hindu muslim sc st obc age mother_yrs_educ father_yrs_educ , a(dt_id) robust ;
	est sto `var' ;
} ;

esttab attended* discuss* member* using "./output/table-effect-of-menarche-age-on-political-participation.tex", 
	replace msign(--)
	drop(height_measure hindu muslim sc st obc age mother_yrs_educ father_yrs_educ _cons) 
	cells(b(label() star fmt(%9.3f %9.3f)) se(par)) 
	star(* 0.10 ** 0.05 *** 0.01) 
	stats(N, fmt(%20s %9.0f) labels("Observations"))
	prehead(\begin{table}[h]\centering \normalsize `"\def\sym#1{\ifmmode^{#1}\else\(^{#1}\)\fi}"'
		\captionsetup{justification=centering} 
		\caption{Effect of Menarche Age on Political Participation, Rural Women}
		\label{table-effect-of-menarche-age-on-political-participation}
		\begin{tabular*}{\hsize}{@{\hskip\tabcolsep\extracolsep\fill}l*{@E}{p{2.8cm}}}
		\toprule)
	legend label  booktabs  collabels( , none)
	mlabels(, depvar span prefix(\multicolumn{@span}{p{3.5cm}}{) suffix(}))
	postfoot(`"\bottomrule"'  \end{tabular*} \captionsetup{justification=justified, width=\hsize} 
		\caption*{\small \textit{Notes:} 
		OLS regression. Data from IHDS 2011--12, rural sample.
		All regressions include district FEs, height, age, parents' education, household caste, religion. 
		Robust SEs in parenthesis. ***\$\,p < 0.01$, **\$\,p < 0.05$, *\$\,p<0.10$.\looseness=-1}
		\vspace{-4ex}
		\end{table}) ;

exit ;
